Jan 25, 2024

If you want to get the SBC requirement WRTD for this class, email me at scott.sutherland@stonybrook.edu and let me know that, your name and SBU ID number. I will give permission to enroll in MAT459 (0 credits).

Below, I typed Pi^27

```
> (Pi)^27
```

$$\pi^{27} \quad (1)$$

```
> evalf(Pi^27); evalf(Pi^27, 20)
```

$$2.648784121 \times 10^{13}$$

$$2.6487841119103630236 \times 10^{13} \quad (2)$$

```
> diff(Pi^27, Pi)
```

Error, invalid input: diff received Pi, which is not valid for its 2nd argument

```
> diff(x^27, x)
```

$$27\, x^{26} \quad (3)$$

Below I am typing diff(pi^27,pi)

```
> diff(pi^27, pi)
```

$$27\, \pi^{26} \quad (4)$$

```
> pi := Pi
```

$$\pi := \pi \quad (5)$$

I can enter commands graphically (this is the default) or so that you can see what I actually typed. To do the latter, type ctrl-m

```
> evalf(pi,30)
```

$$3.14159265358979323846264338328 \quad (6)$$

```
> evalf(pi, 30)
```

$$3.14159265358979323846264338328 \quad (7)$$

```
> evalf(e)
```

$$e \quad (8)$$

```
> evalf(E)
```

$$E \quad (9)$$

```
> E := exp(1);
```

$$E := e \quad (10)$$

```
> evalf(E);
```

$$2.718281828 \quad (11)$$

```
>
```

1. Plot the function f(x)=2 sin(x) - x^3 -1/5 for -4 < x < 4.
Find all the zeros of the function correct to 20 digits.

hint: Digits and fsolve might be useful.

Here is doing stuff wrong, but a way I can make work

```
> f := 2 sin(x) - x^3 - 1/5
```

$$f := 2\sin(x) - x^3 - \frac{1}{5} \tag{1.1}$$

```
> f(x);
```

$$2\sin(x)(x) - x(x)^3 - \frac{1}{5} \tag{1.2}$$

That's not what I meant!
But it kinda works:

```
> diff(f, x);
```

$$2\cos(x) - 3x^2 \tag{1.3}$$

```
> eval(f, x = Pi/6);
```

$$\frac{4}{5} - \frac{\pi^3}{216} \tag{1.4}$$

```
> eval(f, x = x^3)
```

$$2\sin(x^3) - x^9 - \frac{1}{5} \tag{1.5}$$

I really meant to write (I'll use g)

```
> g(x) := 2 sin(x) - x^3 - 1/5
```

$$g := x \mapsto 2\cdot\sin(x) - x^3 - \frac{1}{5} \tag{1.6}$$

or

```
> h := x->2 sin(x) - x^3 - 1/5
```

$$h := x \mapsto 2\cdot\sin(x) - x^3 - \frac{1}{5} \tag{1.7}$$

```
> h(x), h(x^3)
```

$$2\sin(x) - x^3 - \frac{1}{5},\ 2\sin(x^3) - x^9 - \frac{1}{5} \tag{1.8}$$

```
> newf := unapply(f, x)
```

$$newf := x \mapsto 2\cdot\sin(x) - x^3 - \frac{1}{5} \tag{1.9}$$

unapply turns an expression into a function.

```
> plot(f, x = -4..4)
```



> $f(3)$

$$2\sin(x)(3) - x(3)^3 - \frac{1}{5} \tag{1.10}$$

Lets go back to the problem, but define f as a function.

> $f(x) := 2\sin(x) - x\text{^}3 - 1/5$

$$f := x \mapsto 2\cdot\sin(x) - x^3 - \frac{1}{5} \tag{12}$$

> $plot(f(x), x = -4..4);$



> $f\left(\frac{\mathrm{Pi}}{6}\right)$

$$\frac{4}{5} - \frac{\pi^3}{216} \tag{13}$$

Since f is a function, the variable isn't specified in the range. f(x) is its value, so we have to say "let x go from -4 to 4"

> $plot(f, -4..4)$



> $plot(f(rabbit), rabbit = -4..4, axes = boxed, color = green, thickness = 2)$



> $solve(f(x) = 0, x)$

$$RootOf\left(5\_Z^3 - 10\sin(\_Z) + 1\right) \quad \textbf{(14)}$$

> $solve(x^3 - 6x^2 - 3 = 0, x)$

$$\frac{\left(76 + 4\sqrt{105}\right)^{1/3}}{2} + \frac{8}{\left(76 + 4\sqrt{105}\right)^{1/3}} + 2, -\frac{\left(76 + 4\sqrt{105}\right)^{1/3}}{4} \quad \textbf{(15)}$$

$$-\frac{4}{\left(76 + 4\sqrt{105}\right)^{1/3}} + 2 + \frac{I\sqrt{3}\left(\frac{\left(76 + 4\sqrt{105}\right)^{1/3}}{2} - \frac{8}{\left(76 + 4\sqrt{105}\right)^{1/3}}\right)}{2},$$

$$-\frac{\left(76 + 4\sqrt{105}\right)^{1/3}}{4} - \frac{4}{\left(76 + 4\sqrt{105}\right)^{1/3}} + 2$$

$$-\frac{I\sqrt{3}\left(\frac{\left(76+4\sqrt{105}\right)^{1/3}}{2}-\frac{8}{\left(76+4\sqrt{105}\right)^{1/3}}\right)}{2}$$

Oh, my what a mess.... lets evauate that as a number to 20 digits.
%means "result of the last command I did"

>

> $evalf(\%, 20)$

$$6.0811247605094980228,\ -0.040562380254749011 4+0.70120216994300630460\ I,\ -0.040562380254749011 4-0.70120216994300630460\ I \quad (16)$$

> $evalf(fsolve(f(x)=0), 20)$

$$0.10068027882300758308 \quad (17)$$

> $r := \%$ # *remember the value of it.*

$$r := 0.10068027882300758308 \quad (18)$$

> $f(r);$

$$0. \quad (19)$$

> $plot\left(f(x), x=-\frac{Pi}{2}..\frac{Pi}{2}\right)$



We can see that there are (at least) three solutions. How do we know there are not more?
Let's use fsolve, as suggested, but we have to figure out how to restrict the place to look for solutions.

> ?*fsolve*

> $fsolve\left(f(x)=0, x=-\frac{Pi}{2}..-\frac{3\ Pi}{8}\right)$

$$-1.2843277 \quad (20)$$

> $s := \%;$

$$s := -1.2843277 \quad (21)$$

> $t := fsolve\left(f(x)=0, x=\frac{3\ Pi}{8}\right)$

$$t := 1.1818052 \quad (22)$$

> $[r, s, t]$

$$[0.10068027882300758308,\ -1.2843277,\ 1.1818052] \quad (23)$$

Oh no, I wanted 20 digits!)

> $evalf\left(fsolve\left(f(x)=0, x=\frac{3\ \mathrm{Pi}}{8}\right), 20\right)$

$$1.1818052229380529855 \quad \textbf{(24)}$$

Alternatively, set `Digits:=20` at the start, and just go for it.

>

> $Digits := 20$

$$Digits := 20 \quad \textbf{(25)}$$

> $sols := \left[fsolve\left(f(x), x=\frac{-3\ \mathrm{Pi}}{8}\right), fsolve(f(x), x=0), fsolve\left(f(x), x=\frac{3\ \mathrm{Pi}}{8}\right)\right]$

$$sols := [-1.2843276757295733392, 0.10068027882300758308, 1.1818052229380529855] \quad \textbf{(26)}$$

> $f(sols[1]), f(sols[2]), f(sols[3])$

$$-1.0 \times 10^{-19}, 0., 0. \quad \textbf{(27)}$$

>